

## 51<sup>th</sup> Fire Department



**What new kind of gameplay is possible after 50 games? Move left, right, up or down.  
Now we need to pressurize a waterhose to the right floor. And again... it fits 1K.**

```
; 51th Fire Department
; Game 51 in 1K hires for the ZX81
```

```
? * TORNADO *
```

```
ORG #4009 ;#4009
DUMP 49161
```

```
fl EQU 6
```

```
basic LD D,#C0 ; preset for 48K bug
JR init0 ; this game has no 48K bug
```

```
DEFB 236,212,28 ; The BASIC
DEFB 126 ; fully placed over sysvar
DEFB 143,0,18 ; start to BASIC=#4009
```

```
eline DEFW last ; needed to load
chadd DEFW last-1
xptr DEFW 0
stkbot DEFW last ; needed to load
stkend DEFW last ; needed to load
berg DEFB 0
mem DEFW 0
DEFB 0 ; 128
```

```

initl      JP    init                ; init can be anywhere

; all above reusable AFTER loading

lastk      DEFB 255,255,255          ; used by ZX81
margin     DEFB 55                   ; used by ZX81
nxtlin     DEFW basic                ; reusable after load

init0      LD     E,L                ; delay intrupts by
          DEFB #26                   ; LD H,64
flagx      DEFB 64                  ; clever setting of flags

          XOR     A                  ; intruptcounter reset
          EX      AF,AF'

taddr      DEFW 0                    ; used by ZX81,no hurting code
          LD      B,4                ; frames is set ok

frames     DEFW #DD*256+1            ; used by ZX81, clever IX set
coprcc     LD      HL,hr              ; set IX
sposn      JR      initl
cdflag     DEFB 64                   ; used by zx81

dead       DEFW 0                    ; filler and variable space

udg1       EQU     room*256/256
udg2       EQU     fire*256/256
udg3       EQU     water*256/256
udg4       EQU     crack*256/256

fire       DEFB 128,30,1,255,164,255,170,72,85
          DEFB 170,144,85,172,156,53,180,138,45
          DEFB 180,73,45,216,105,27,208,81,11
          DEFB 210,70,75,229,42,199,212,227,75
          DEFB 210,1,75,169,255,149,255,255,255

room       DEFB 128,0,1,255,255,255,170,24,85
          DEFB 170,36,85,172,60,53,180,0,45
          DEFB 180,0,45,216,168,27,208,82,11
          DEFB 208,37,11,224,162,7,208,239,11
          DEFB 208,126,11,168,60,21,255,255,255

crack      DEFB 128,0,1,255,255,255,170,2,85
          DEFB 170,130,85,172,122,53,180,38,45
          DEFB 180,32,45,216,22,27,208,72,11
          DEFB 208,100,11,224,82,7,208,74,11
          DEFB 208,70,11,168,64,21,255,255,255
          DEFB 128,0,1

water      DEFB 128,85,1,255,85,255,170,42,85
          DEFB 170,42,85,172,42,53,180,42,45
          DEFB 180,85,45,216,85,27,208,85,11
          DEFB 208,85,11,224,42,7,208,42,11
          DEFB 208,42,11,168,42,21,255,85,127
          DEFB 128,85,1

hr         LD      HL,lowres+#8000    ; the lowres display
          LD      BC,#241             ; minimum needed
          LD      A,#1E
          LD      I,A
          LD      A,#FB
          CALL    #2B5

          LD      B,4                  ; sync hires display

```

```

hr00      DJNZ hr00
          LD      A,(HL)

          LD      HL,low      ; return from display
          EXX

          LD      A,#40
          LD      I,A        ; set high byte display

          LD      B,fl+1      ; the number of floors +1
          LD      (saveSP+1),SP ; save SP
          LD      SP,ixpoint  ; get floors from stack
          JR      cloop

loop1     NOP
          POP     IX          ; get next floor
          LD      HL,udg1*256+udg2 ; window and fire
          LD      DE,udg3*256+udg4 ; water and cracked window
          LD      C,16        ; 16 lines per floor
          JR      nline

bloop     INC     DE          ; next line cracked window
          INC     E
          INC     E

          INC     D          ; next line water
          INC     D
          INC     D

          INC     HL          ; next line fire
          INC     HL
          INC     HL

          INC     H          ; next line window
          INC     H
          INC     H

nline     DEC     C
          LD      A,(HL)      ; filler
          JP      (IX)        ; show floor

low       EXX
          JP      Z,cloop     ; show next floor
          JP      bloop       ; show rest current floor

cloop     XOR     A          ; extra line

          POP     HL          ; filler
          PUSH    HL
          DEC     HL
          EX      (SP),HL
          EX      (SP),HL

          LD      HL,low2
          JP      lbuf2+#8000

low2      LD      A,(HL)
          DEC     HL
          DJNZ    loop1

saveSP    LD      SP,0        ; retrieve stack

          LD      B,6
hr01      LD      DE,truckudg

```

```

NOP
DJNZ hr01

LD A,dataline/256
LD I,A

LD BC,#BFF

truckpos LD A,dataline*256/256
EX DE,HL
truckx LD DE,dataline
LDI
LDI
LDI
EX DE,HL
LD HL,low3
JP lbuf2+#8000

low3 DJNZ truckpos

CALL #292 ; back from intrupt
CALL #220
LD IX,hr
JP #2A4

truckudg DEFB 61,255,255,69,23,19,69,119,117
DEFB 253,55,21,253,215,115,255,53,255
DEFB 215,255,235,171,255,213,16,0,8
DEFB 0 ; needed for ok display

dataline DEFW 0,0,0,0,0,0,0,0,0,0,0,0

lbuf2 LD R,A
DEFW 0,0,0,0,0,0,0,0,0,0,0,0
JP (HL)

ixpoint DEFW lbuf+32768 ; each floor
DEFW lbuf+32794 ; has own display
DEFW lbuf+32820 ; 6 floor building
DEFW lbuf+32846
DEFW lbuf+32872
DEFW lbuf+32898

flsize EQU 26*fl-26

eog LD HL,score-1
LD DE,hiscore-1
LD C,7 ; b=0 here

fihi INC HL
INC DE
DEC C
JR Z,start
LD A,(DE)
CP (HL)
JR Z,fihi
CALL C,#19F9 ; set new high

start LD A,(lastk) ; game over, wait for
SUB %10111111 ; newline
JR NZ,start

LD L,score*256/256

```

```

ressc    LD    (HL),28            ; reset score
          INC   HL
          CP    (HL)
          JR    NZ,ressc

          LD    L,lbuf*256/256+24+3
          LD    DE,65535-5        ; -6
          LD    BC,1024+flsize
mkcopy   ADD    HL,DE
          LD    (HL),#3C          ; repair all rooms
          DJNZ  mkcopy

          LD    L,lbuf*256/256    ; all floors the same
          LD    DE,init
          LDIR

          CALL  showall          ; show floors

nextroom LD    B,fl
          CALL  rnd
          LD    C,A              ; set floor

          CALL  rnd
          AND   3
          INC   A
          LD    B,A              ; set app on floor
          LD    (fircol+1),A      ; save firecolumn

          CALL  field
          CP    #7B              ; burnt out test
          JR    Z,nextroom

; 7a water, 7b broken, 7c room, 7d fire
          LD    (HL),#7D          ; set room on fire
          LD    (fireapp+1),HL    ; save app location

          LD    B,20
          CALL  rnd
          ADD   A,B
          ADD   A,B
          LD    (dead+1),A        ; set timer

playloop JR    Z,eog

          LD    A,%11011111      ; Y-P
          IN    A,(254)
          LD    HL,truckx+1
          LD    C,(HL)
          RRA                    ; P
          JR    C,t2
          INC   (HL)

t2        RRA                    ; O
          JR    C,t3
          DEC   (HL)

t3        LD    A,(HL)
          SUB   dataline*256/256
          CP    19
          JR    C,colcheck        ; not out of line
          LD    (HL),C            ; undo illegal move

colcheck LD    A,(HL)            ; get xpos
fircol   LD    B,0
          LD    C,B
          SUB   dataline*256/256-6

```

```

colnr      SUB 6
           DJNZ colnr
           JR  NZ,delaylp      ; not in column

           LD  B,C
           LD  A,(lastk)
           LD  HL,power
           SUB %11111011      ; Q-T
           JR  Z,release

           CP  %11111101-%11111011 ; A-G
           JR  NZ,drive

old         CP  0              ; keyrepeat, is no pumping
           LD  B,A
           JR  Z,setold

           LD  A,(HL)
           SUB 28+fl
           JR  Z,setold
           INC (HL)            ; increase waterpressure

setold      LD  A,B
drive       LD  (old+1),A      ; signal other key pressed

delaylp     CALL #4016         ; wait some time

           LD  HL,dead+1      ; decrease timer
           DEC (HL)
           JR  NZ,pl
           DEC HL
           INC (HL)            ; add a missed window
           LD  A,(HL)
           AND 7
           LD  HL,(fireapp+1)
           INC HL
           INC HL
           INC HL
           LD  (HL),#3B        ; set signal broken window
           PUSH AF
           CALL showwall       ; show windows
           POP AF
           JR  NZ,nr           ; not eog, get next fire

pl          JP  playloop       ; continue test eog

release     LD  C,A            ; C now 0
           LD  A,(HL)
           SUB 28
           JR  Z,delaylp

showwater   CALL field         ; get lowest field
           LD  D,(HL)          ; read what is it
           LD  (HL),#7A        ; show water
           LD  A,(power)
           SUB C
           INC C
           CP  28
           JR  NZ,showwater    ; do full pressure

           PUSH DE              ; save latest apartment

           LD  A,256-6
           CALL #4018

```

```

CALL showall

fireapp    LD    HL,0
           LD    (HL),#7D          ; set fire back on
           POP   AF                ; test fire in app
           CP    #7D              ; fire extinguished?
           JR    NZ,drive

           LD    (HL),#7C          ; erase fire

; score = floor * remaining time
addc        LD    A,(dead+1)      ; get timer
           LD    B,A
addb        LD    HL,score+6
           DEFB 17
ten         LD    (HL),28
           DEC   HL
           INC   (HL)
           LD    A,(HL)
           CP    38
           JR    Z,ten
           DJNZ  addb
           DEC   C
           JR    NZ,addc

nr          JP    nextroom        ; set next fire

showall     LD    B,24            ; show all original fields
           LD    HL,lbuf
allfield    LD    D,H            ; destination
           LD    E,L
           INC   HL
           INC   HL
           INC   HL
           LD    A,(HL)          ; get save location
           SET   6,A             ; make it LD A,r
           LD    (DE),A
aok         INC   HL
hl2         INC   HL
           INC   HL
           LD    A,(HL)          ; test end of line
           CP    #D9
           JR    Z,hl2
           DJNZ  allfield
           LD    L,power*256/256 ; reset pressure
           LD    (HL),28
           RET

rnd         LD    HL,(frames)     ; read RND from ROM
rseed       LD    DE,0
           ADD   HL,DE
           DEC   HL
           LD    A,H
           AND   #1F
           LD    H,A
           LD    (rseed+1),HL
           LD    A,(HL)
frnd        SUB   B
           JR    NC,frnd
           ADC   A,B
           RET

field       LD    HL,lbuf-6-26

```

```

        LD    A,fl+1
        SUB   C
flbuf    LD    DE,26
        ADD   HL,DE
        DEC   A
        JR    NZ,flbuf
        LD    E,6
        LD    A,B
ffield   ADD   HL,DE
        DEC   A
        JR    NZ,ffield
        LD    A,(HL)
        RET

x        EQU   101

lowres   DEFB  118
score    DEFB  28,28,28,28,28,28,0

        DEFB  "F"+x,"I"+x,"R"+x,"E"+x,"F"+x,"I"+x
        DEFB  "G"+x,"H"+x,"T"+x,"E"+x,"R"+x,0
power    DEFB  28,0
hiscore  DEFB  28,28,28,28,33,29
        DEFB  118

scsize   EQU   fl*26

space    EQU   #4400-26-scsize-$

        DEFS  space                ; no space left

lbuf     LD    A,E
        LD    R,A
        DEFB  0,0,0

        LD    A,E
        LD    R,A
        DEFB  0,0,0

        LD    A,E
        LD    R,A
        DEFB  0,0,0

        LD    A,E
        LD    R,A
        DEFB  0,0,0
        EXX
        JP    (HL)

init     LD    SP,#4400
        LDIR                    ; repair 48K bug, DEFAULT

        LD    DE,#4016           ; delay routine over sysvar
        LD    HL,delay
        LD    C,10
        LDIR

        LD    HL,#4015           ; built floor line
        LD    (HL),252
        DEC   L
        LD    (HL),255
        JR    NZ,mkline         ; built horizontal line

mkline   LD    HL,eog            ; start on end of game

```



```

        PUSH HL
        LD HL,lbud
        LD DE,init
        LD BC,26*fl-26      ; make buildingfloors
        JP #19F9

delay   LD A,256-3
        LD HL,frames
        ADD A,(HL)
wfr     CP (HL)
        JR NZ,wfr
        RET

vars    DEFB 128
?
last    EQU $

```